
F11 – Sparse Linear Algebra

F11BDF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F11BDF is a set-up routine, the first in a suite of three routines for the iterative solution of a real general
(nonsymmetric) system of simultaneous linear equations. F11BDF must be called before F11BEF, the
iterative solver. The third routine in the suite, F11BFF, can be used to return additional information
about the computation.

These three routines are suitable for the solution of large sparse general (nonsymmetric) systems of
equations.

2 Specification

SUBROUTINE F11BDF(METHOD, PRECON, NORM, WEIGHT, ITERM, N, M, TOL,
1 MAXITN, ANORM, SIGMAX, MONIT, LWREQ, WORK,
2 LWORK, IFAIL)
INTEGER ITERM, N, M, MAXITN, MONIT, LWREQ, LWORK, IFAIL
real TOL, ANORM, SIGMAX
CHARACTER∗1 PRECON, NORM, WEIGHT
CHARACTER∗(*) METHOD
real WORK(LWORK)

3 Description

The suite consisting of the routines F11BDF, F11BEF, F11BFF is designed to solve the general
(nonsymmetric) system of simultaneous linear equations Ax = b of order n, where n is large and the
coefficient matrix A is sparse.

F11BDF is a set-up routine which must be called before F11BEF, the iterative solver. The third routine
in the suite, F11BFF, can be used to return additional information about the computation. A choice of
methods is available:

restarted generalized minimum residual method (RGMRES);

conjugate gradient squared method (CGS);

bi-conjugate gradient stabilized (�) method (Bi-CGSTAB (�));

transpose-free quasi-minimal residual method (TFQMR).

3.1 Restarted Generalized Minimum Residual Method (RGMRES)

The restarted generalized minimum residual method (RGMRES) (Saad and Schultz [4], Barrett et al. [2],
Dias da Cunha and Hopkins [3]) starts from the residual r0 = b−Ax0, where x0 is an initial estimate for
the solution (often x0 = 0). An orthogonal basis for the Krylov subspace span{Akr0}, for k = 0, 1, 2, . . ., is
generated explicitly: this is referred to as Arnoldi’s method (Arnoldi [8]). The solution is then expanded
onto the orthogonal basis so as to minimize the residual norm ‖b − Ax‖2. The lack of symmetry of
A implies that the orthogonal basis is generated by applying a ‘long’ recurrence relation, whose length
increases linearly with the iteration count. For all but the most trivial problems, computational and
storage costs can quickly become prohibitive as the iteration count increases. RGMRES limits these
costs by employing a restart strategy: every m iterations at most, the Arnoldi process is restarted from
rl = b−Axl, where the subscript l denotes the last available iterate. Each group of m iterations is referred
to as a ‘super-iteration’. The value of m is chosen in advance and is fixed throughout the computation.
Unfortunately, an optimum value of m cannot easily be predicted.

3.2 Conjugate Gradient Squared Method (CGS)

The conjugate gradient squared method (CGS) (Sonneveld [5], Barrett et al. [2], Dias da Cunha and
Hopkins [3]) is a development of the bi-conjugate gradient method where the nonsymmetric Lanczos
method is applied to reduce the coefficients matrix to real tridiagonal form: two bi-orthogonal sequences
of vectors are generated starting from the residual r0 = b − Ax0, where x0 is an initial estimate for
the solution (often x0 = 0) and from the shadow residual r̂0 corresponding to the arbitrary problem
AT x̂ = b̂, where b̂ can be any vector, but in practice is chosen so that r0 = r̂0. In the course
of the iteration, the residual and shadow residual ri = Pi(A)r0 and r̂i = Pi(A

T )r̂0 are generated,
where Pi is a polynomial of order i, and bi-orthogonality is exploited by computing the vector product
ρi = (r̂i, ri) = (Pi(A

T )r̂0, Pi(A)r0) = (r̂0, P
2
i (A)r0). Applying the ‘contraction’ operator Pi(A) twice, the

iteration coefficients can still be recovered without advancing the solution of the shadow problem, which
is of no interest. The CGS method often provides fast convergence; however, there is no reason why the
contraction operator should also reduce the once reduced vector Pi(A)r0: this may well lead to a highly
irregular convergence which may result in large cancellation errors.

3.3 Bi-Conjugate Gradient Stabilized (�) Method (Bi-CGSTAB (�))

The bi-conjugate gradient stabilized (�) method (Bi-CGSTAB (�)) (van der Vorst [6], Sleijpen and
Fokkema [7], Dias da Cunha and Hopkins [3]) is similar to the CGS method above. However, instead
of generating the sequence {P 2

i (A)r0}, it generates the sequence {Qi(A)Pi(A)r0} where the Qi(A) are
polynomials chosen to minimize the residual after the application of the contraction operator Pi(A). Two
main steps can be identified for each iteration: an OR (Orthogonal Residuals) step where a basis of order �
is generated by a Bi-CG iteration and an MR (Minimum Residuals) step where the residual is minimized
over the basis generated, by a method akin to GMRES. For � = 1, the method corresponds to the
Bi-CGSTAB method of van der Vorst [6]. For � > 1, more information about complex eigenvalues of the
iteration matrix can be taken into account, and this may lead to improved convergence and robustness.
However, as � increases, numerical instabilities may arise. For this reason, a maximum value of � = 10 is
imposed, but probably � = 4 is sufficient in most cases.

3.4 Transpose-Free Quasi-Minimal Residual Method (TFQMR)

The transpose-free quasi-minimal residual method (TFQMR) (Freund and Nachtigal [10], Freund [11])
is conceptually derived from the CGS method. The residual is minimised over the space of the residual
vectors generated by the CGS iterations under the simplifying assumption that residuals are almost
orthogonal. In practice, this is not the case but theoretical analysis has proved the validity of the method.
This has the effect of remedying the rather irregular convergence behaviour with wild oscillations in the
residual norm that can degrade the numerical performance and robustness of the CGS method. In general,
the TFQMR method can be expected to converge at least as fast as the CGS method, in terms of number
of iterations, although each iteration involves a higher operation count. When the CGS method exhibits
irregular convergence, the TFQMR method can produce much smoother, almost monotonic convergence
curves. However, the close relationship between the CGS and TFQMR method implies that the overall
speed of convergence is similar for both methods. In some cases, the TFQMR method may converge
faster than the CGS method.

3.5 General Considerations

For each method, a sequence of solution iterates {xi} is generated such that, hopefully, the sequence of
the residual norms {‖ri‖} converges to a required tolerance. Note that, in general, convergence, when it
occurs, is not monotonic.

In the RGMRES and Bi-CGSTAB (�) methods above, the user’s program must provide the maximum
number of basis vectors used, m or �, respectively; however, a smaller number of basis vectors may be
generated and used when the stability of the solution process requires this (see Section 8).

Faster convergence can be achieved using a preconditioner (Golub and Van Loan [1], Barrett et al. [2]).
A preconditioner maps the original system of equations onto a different system, say

Āx̄ = b̄, (1)

with, hopefully, better characteristics with respect to its speed of convergence: for example, the condition
number of the coefficients matrix can be improved or eigenvalues in its spectrum can be made to coalesce.

An orthogonal basis for the Krylov subspace span{Ākr̄0}, for k = 0, 1, . . ., is generated and the solution
proceeds as outlined above. The algorithms used are such that the solution and residual iterates of
the original system are produced, not their preconditioned counterparts. Note that an unsuitable
preconditioner or no preconditioning at all may result in a very slow rate, or lack, of convergence.
However, preconditioning involves a trade-off between the reduction in the number of iterations required
for convergence and the additional computational costs per iteration. Also, setting up a preconditioner
may involve non-negligible overheads.

A left preconditioner M−1 can be used by the RGMRES, CGS and TFQMR methods, such that
Ā = M−1A ∼ In in (1), where In is the identity matrix of order n; a right preconditioner M−1 can
be used by the Bi-CGSTAB (�) method, such that Ā = AM−1 ∼ In. These are formal definitions, used
only in the design of the algorithms; in practice, only the means to compute the matrix–vector products
v = Au and v = AT u (the latter only being required when an estimate of ‖A‖1 or ‖A‖∞ is computed
internally), and to solve the preconditioning equations Mv = u are required, i.e., explicit information
about M , or its inverse is not required at any stage.

The first termination criterion
‖rk‖p ≤ τ (‖b‖p + ‖A‖p ‖xk‖p) (2)

is also available for all three methods. In (2), p = 1, ∞ or 2 and τ denotes a user-specified tolerance
subject to max(10,

√
n), ε ≤ τ < 1, where ε is the machine precision. Facilities are provided for the

estimation of the norm of the coefficients matrix ‖A‖1 or ‖A‖∞, when this is not known in advance, by
applying Higham’s method (Higham [9]). Note that ‖A‖2 cannot be estimated internally. This criterion
uses an error bound derived from backward error analysis to ensure that the computed solution is the
exact solution of a problem as close to the original as the termination tolerance requires. Termination
criteria employing bounds derived from forward error analysis are not used because any such criteria
would require information about the condition number κ(A) which is not easily obtainable.

The second termination criterion

‖r̄k‖2 ≤ τ (‖r̄0‖2 + σ1(Ā) ‖∆x̄k‖2) (3)

is available for all methods except TFQMR. In (3), σ1(Ā) = ‖Ā‖2 is the largest singular value of the
(preconditioned) iteration matrix Ā. This termination criterion monitors the progress of the solution
of the preconditioned system of equations and is less expensive to apply than criterion (2) for the Bi-
CGSTAB (�) method with � > 1. Only the RGMRES method provides facilities to estimate σ1(Ā)
internally, when this is not supplied (see Section 8).

Termination criterion (2) is the recommended choice, despite its additional costs per iteration when using
the Bi-CGSTAB (�) method with � > 1. Also, if the norm of the initial estimate is much larger than the
norm of the solution, that is, if ‖x0‖ 
 ‖x‖, a dramatic loss of significant digits could result in complete
lack of convergence. The use of criterion (2) will enable the detection of such a situation, and the iteration
will be restarted at a suitable point. No such restart facilities are provided for criterion (3).

Optionally, a vector w of user-specified weights can be used in the computation of the vector norms in
termination criterion (2), i.e., ‖v‖(w)

p = ‖v(w)‖p, where (v
(w))i = wi vi, for i = 1, 2, . . . , n. Note that the

use of weights increases the computational costs.

The sequence of calls to the routines comprising the suite is enforced: first, the set-up routine F11BDF
must be called, followed by the solver F11BEF. F11BFF can be called either when F11BEF is carrying
out a monitoring step or after F11BEF has completed its tasks. Incorrect sequencing will raise an error
condition.

In general, it is not possible to recommend one method in preference to another. RGMRES is often used
in the solution of systems arising from PDEs. On the other hand, it can easily stagnate when the size m
of the orthogonal basis is too small, or the preconditioner is not good enough. CGS can be the fastest
method, but the computed residuals can exhibit instability which may greatly affect the convergence
and quality of the solution. Bi-CGSTAB (�) seems robust and reliable, but it can be slower than the
other methods: if a preconditioner is used and � > 1, Bi-CGSTAB (�) computes the solution of the
preconditioned system x̄k = Mxk: the preconditioning equations must be solved to obtain the required
solution. The algorithm employed limits to 10% or less, when no intermediate monitoring is requested, the
number of times the preconditioner has to be thus applied compared with the total number of applications
of the preconditioner. TFQMR can be viewed as a more robust variant of the CGS method: it shares the

CGS method speed but avoids the CGS fluctuations in the residual, which may give rise to instability.
Also, when the termination criterion (2) is used, the CGS, Bi-CGSTAB (�) and TFQMR methods will
restart the iteration automatically when necessary in order to solve the given problem.

5 Parameters

1: METHOD — CHARACTER*(*) Input

On entry: the iterative method to be used. The possible choices are:

’RGMRES’ restarted generalized minimum residual method;
’CGS’ conjugate gradient squared method;
’BICGSTAB’ bi-conjugate gradient stabilized (�) method;
’TFQMR’ transpose-free quasi-minimal residual method.

Constraint: METHOD = ’RGMRES’, ’CGS’, ’BICGSTAB’ or ’TFQMR’.

2: PRECON — CHARACTER*1 Input

On entry: determines whether preconditioning is used. The possible choices are:

’N’ no preconditioning;
’P’ preconditioning.

Constraint: PRECON = ’N’ or ’P’.

3: NORM — CHARACTER*1 Input

On entry: defines the matrix and vector norm to be used inthe termination criteria. The possible
choices are:

’1’ l1 norm;
’I’ l∞ norm;
’2’ l2 norm.

Suggested value: NORM = ’I’, if ITERM = 1; NORM = ’2’, if ITERM = 2.

Constraints: if ITERM = 1, then NORM = ’1’, ’I’ or ’2’; if ITERM = 2, then NORM = ’2’.

4: WEIGHT — CHARACTER*1 Input

On entry: specifies whether a vector w of user-supplied weights is to be used in the computation
of the vector norms required in termination criterion (2) (ITERM = 1): ‖v‖(w)

p = ‖v(w)‖p, where
v
(w)
i = wi vi, for i = 1, 2, . . . , n. The suffix p = 1, 2,∞ denotes the vector norm used, as specified
by the parameter NORM. Note that weights cannot be used when ITERM = 2, i.e., when criterion
(3) is used. The possible choices are:

’W’ user-supplied weights are to be used and must be supplied on initial entry to F11BEF.
’N’ all weights are implicitly set equal to one. Weights do not need to be supplied on initial entry

to F11BEF.

Suggested value: WEIGHT = ’N’.

Constraints: if ITERM = 1, then WEIGHT = ’W’ or ’N’; if ITERM = 2, then WEIGHT = ’N’.

5: ITERM — INTEGER Input

On entry: defines the termination criterion to be used. The possible choices are:

1 use the termination criterion defined in (2);
2 use the termination criterion defined in (3).

Suggested value: ITERM = 1.

Constraints: if METHOD = ’RGMRES’, ’CGS’ or ’BICGSTAB’, ITERM = 1 or 2;

if METHOD = ’TFQMR’, ITERM = 1.

6: N — INTEGER Input

On entry: the order n of the matrix A.

Constraint: N > 0.

7: M — INTEGER Input

On entry: if METHOD = ’RGMRES’, M is the dimension m of the restart subspace.

If METHOD = ’BICGSTAB’, M is the order � of the polynomial Bi-CGSTAB method;

otherwise, M is not referenced.

Constraints:

if METHOD = ’RGMRES’, 0 < M ≤ min(N,50);
if METHOD = ’BICGSTAB’, 0 < M ≤ min(N,10).

8: TOL — real Input

On entry: the tolerance τ for the termination criterion. If TOL ≤ 0.0, τ = max(
√

ε,
√

n ε) is used,
where ε is the machine precision. Otherwise τ = max(TOL, 10ε,

√
n ε) is used.

Constraint: TOL < 1.0.

9: MAXITN — INTEGER Input

On entry: the maximum number of iterations.

Constraint: MAXITN > 0.

10: ANORM — real Input

On entry: if ANORM > 0.0, the value of ‖A‖p to be used in the termination criterion (2) (ITERM
= 1).

If ANORM ≤ 0.0, ITERM = 1 and NORM = ’1’ or ’I’, then ‖A‖1 = ‖A‖∞ is estimated internally
by F11BEF.

If ITERM = 2, then ANORM is not referenced.

Constraint: if ITERM = 1 and NORM = ’2’, then ANORM > 0.0.

11: SIGMAX — real Input

On entry: if ITERM = 2, the largest singular value σ1 of the preconditioned iteration matrix;
otherwise, SIGMAX is not referenced.

If SIGMAX ≤ 0.0, ITERM = 2 and METHOD = ’RGMRES’, then the value of σ1 will be estimated
internally.

Constraint: if METHOD = ’CGS’ or ’BICGSTAB’ and ITERM = 2, then SIGMAX > 0.0.

12: MONIT — INTEGER Input

On entry: if MONIT > 0, the frequency at which a monitoring step is executed by F11BEF: if
METHOD = ’CGS’ or ’TFQMR’, a monitoring step is executed every MONIT iterations; otherwise,
a monitoring step is executed every MONIT super-iterations (groups of up to m or � iterations for
RGMRES or Bi-CGSTAB (�), respectively).

There are some additional computational costs involved in monitoring the solution and residual
vectors when the Bi-CGSTAB (�) method is used with � > 1.

Constraint: MONIT ≤ MAXITN.

13: LWREQ — INTEGER Output

On exit: the minimum amount of workspace required by F11BEF. (See also Section 5 of the
document for F11BEF.)

14: WORK (LWORK) — real array Output

On exit: the workspace WORK is initialised by F11BDF. It must not be modified before calling the
next routine in the suite, namely F11BEF.

15: LWORK — INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F11BDF
was called.

Constraint: LWORK ≥ 100.

Note: although the minimum value of LWORK ensures the correct functioning of F11BDF, a larger
value is required by the other routines in the suite, namely F11BEF and F11BFF. The required
value is as follows:

Method Requirements

RGMRES LWORK = 100 + n(m + 3) + m(m + 5) + 1, where m is the dimension of the
basis;

CGS LWORK = 100 + 7n;
Bi-CGSTAB (�) LWORK = 100 + (2n+ �)(�+ 2) + p, where � is the order of the method;
TFQMR LWORK = 100 + 10n.

where

p = 2n if � > 1 and ITERM = 2 was supplied;
p = n if � > 1 and a preconditioner is used or ITERM = 2 was supplied;
p = 0 otherwise.

16: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Errors and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors detected by the routine:

IFAIL = −i

On entry, the ith argument had an illegal value.

IFAIL = 1

F11BDF has been called out of sequence.

7 Accuracy

Not applicable.

8 Further Comments

RGMRES can estimate internally the maximum singular value σ1 of the iteration matrix, using σ1 ∼ ‖T ‖1

where T is the upper triangular matrix obtained by QR factorization of the upper Hessenberg matrix
generated by the Arnoldi process. The computational costs of this computation are negligible when
compared to the overall costs.

Loss of orthogonality in the RGMRES method, or of bi-orthogonality in the Bi-CGSTAB (�) method
may degrade the solution and speed of convergence. For both methods, the algorithms employed include
checks on the basis vectors so that the number of basis vectors used for a given super-iteration may be
less than the value specified in the input parameter M. Also, the CGS, Bi-CGSTAB (�) and TFQMR
methods will restart automatically the computation from the last available iterates, when the stability of
the solution process requires it.

Termination criterion (3), when available, involves only the residual (or norm of the residual) produced
directly by the iteration process: this may differ from the norm of the true residual r̃k = b − Axk,
particularly when the norm of the residual is very small. Also, if the norm of the initial estimate of the
solution is much larger than the norm of the exact solution, convergence can be achieved despite very
large errors in the solution. On the other hand, termination criterion (3) is cheaper to use and inspects
the progress of the actual iteration. Termination criterion (2) should be preferred in most cases, despite
its slightly larger costs.

9 Example

The example solves an 8× 8 nonsymmetric system of simultaneous linear equations using the bi-conjugate
gradient stabilized method of order � = 2, where the coefficients matrix A has a random sparsity pattern.
An incomplete LU preconditioner is used (routines F11DAF and F11DBF).

9.1 Example Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F11BDF Example Program Text.
* Mark 19 Release. NAG Copyright 1999.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, LA, LIWORK, LWORK
PARAMETER (NMAX=1000,LA=10000,LIWORK=7*NMAX+2,LWORK=6*NMAX)

* .. Local Scalars ..
real ANORM, DTOL, SIGMAX, STPLHS, STPRHS, TOL
INTEGER I, IFAIL, IFAIL1, IFAILX, IREVCM, ITERM, ITN,

+ LFILL, LWREQ, M, MAXITN, MONIT, N, NNZ, NNZC,
+ NPIVM
LOGICAL LOOP
CHARACTER MILU, NORM, PRECON, PSTRAT, WEIGHT
CHARACTER*8 METHOD

* .. Local Arrays ..
real A(LA), B(NMAX), WGT(NMAX), WORK(LWORK), X(NMAX)
INTEGER ICOL(LA), IDIAG(NMAX), IPIVP(NMAX), IPIVQ(NMAX),

+ IROW(LA), ISTR(NMAX+1), IWORK(LIWORK)
* .. External Subroutines ..

EXTERNAL F11BDF, F11BEF, F11BFF, F11DAF, F11DBF, F11XAF
* .. Executable Statements ..

WRITE (NOUT,*) ’F11BDF Example Program Results’
*
* Skip heading in data file
*

READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read or initialize the parameters for the iterative solver
*

READ (NIN,*) METHOD
READ (NIN,*) PRECON, NORM, WEIGHT, ITERM
READ (NIN,*) M, TOL, MAXITN
READ (NIN,*) MONIT
ANORM = 0.0e0
SIGMAX = 0.0e0

*
* Read the parameters for the preconditioner
*

READ (NIN,*) LFILL, DTOL
READ (NIN,*) MILU, PSTRAT

*
* Read the number of non-zero elements of the matrix A, then read
* the non-zero elements
*

READ (NIN,*) NNZ
DO 20 I = 1, NNZ

READ (NIN,*) A(I), IROW(I), ICOL(I)
20 CONTINUE

*
* Read right-hand side vector b and initial approximate solution
*

READ (NIN,*) (B(I),I=1,N)
READ (NIN,*) (X(I),I=1,N)

*
* Calculate incomplete LU factorization
*

IFAIL = 0
MILU = ’N’
CALL F11DAF(N,NNZ,A,LA,IROW,ICOL,LFILL,DTOL,PSTRAT,MILU,IPIVP,

+ IPIVQ,ISTR,IDIAG,NNZC,NPIVM,IWORK,LIWORK,IFAIL)
*
* Call F11BDF to initialize the solver
*

IFAIL = 0
CALL F11BDF(METHOD,PRECON,NORM,WEIGHT,ITERM,N,M,TOL,MAXITN,

+ ANORM,SIGMAX,MONIT,LWREQ,WORK,LWORK,IFAIL)
*
* Call repeatedly F11BEF to solve the equations
* Note that the arrays B and X are overwritten
*
* On final exit, X will contain the solution and B the residual
* vector
*

IFAIL = 0
IREVCM = 0
LOOP = .TRUE.

*
LWREQ = LWORK

40 CONTINUE
CALL F11BEF(IREVCM,X,B,WGT,WORK,LWREQ,IFAIL)
IF (IREVCM.EQ.-1) THEN

IFAILX = -1
CALL F11XAF(’Transpose’,N,NNZ,A,IROW,ICOL,’No checking’,X,B,

+ IFAILX)
ELSE IF (IREVCM.EQ.1) THEN

IFAILX = -1
CALL F11XAF(’No transpose’,N,NNZ,A,IROW,ICOL,’No checking’,

+ X,B,IFAILX)
ELSE IF (IREVCM.EQ.2) THEN

IFAIL1 = -1
CALL F11DBF(’No transpose’,N,A,LA,IROW,ICOL,IPIVP,IPIVQ,

+ ISTR,IDIAG,’No checking’,X,B,IFAIL1)
IF (IFAIL1.NE.0) IREVCM = 6

ELSE IF (IREVCM.EQ.3) THEN
IFAIL1 = 0
CALL F11BFF(ITN,STPLHS,STPRHS,ANORM,SIGMAX,WORK,LWREQ,

+ IFAIL1)
WRITE (NOUT,99999) ITN, STPLHS
WRITE (NOUT,99998)
WRITE (NOUT,99997) (X(I),B(I),I=1,N)

ELSE IF (IREVCM.EQ.4) THEN
LOOP = .FALSE.

END IF
IF (LOOP) GO TO 40

*
* Obtain information about the computation
*

IFAIL1 = 0
CALL F11BFF(ITN,STPLHS,STPRHS,ANORM,SIGMAX,WORK,LWREQ,IFAIL1)

*
* Print the output data
*

WRITE (NOUT,99996)
WRITE (NOUT,99995) ’Number of iterations for convergence: ’,

+ ITN
WRITE (NOUT,99994) ’Residual norm: ’,

+ STPLHS
WRITE (NOUT,99994) ’Right-hand side of termination criterion:’,

+ STPRHS
WRITE (NOUT,99994) ’1-norm of matrix A: ’,

+ ANORM
*
* Output x
*

WRITE (NOUT,99998)
WRITE (NOUT,99997) (X(I),B(I),I=1,N)

END IF
STOP

*
99999 FORMAT (/1X,’Monitoring at iteration no.’,I4,/1X,1P,’residual no’,

+ ’rm: ’,e14.4)
99998 FORMAT (/2X,’ Solution vector’,2X,’ Residual vector’)
99997 FORMAT (1X,1P,e16.4,1X,e16.4)
99996 FORMAT (/1X,’Final Results’)
99995 FORMAT (1X,A,I4)
99994 FORMAT (1X,A,1P,e14.4)

END

9.2 Example Data

F11BDF Example Program Data
8 N
’BICGSTAB’ METHOD
’P’ ’1’ ’N’ 1 PRECON, NORM, WEIGHT, ITERM
1 1.0E-8 20 M, TOL, MAXITN
1 MONIT
0 0.0 LFILL, DTOL
’N’ ’C’ MILU, PSTRAT

24 NNZ
2. 1 1

-1. 1 4
1. 1 8
4. 2 1

-3. 2 2
2. 2 5

-7. 3 3
2. 3 6
3. 4 1

-4. 4 3
5. 4 4
5. 4 7

-1. 5 2
8. 5 5

-3. 5 7
-6. 6 1
5. 6 3
2. 6 6

-5. 7 3
-1. 7 5
6. 7 7

-1. 8 2
2. 8 6
3. 8 8 A(I), IROW(I), ICOL(I), I=1,...,NNZ
6. 8. -9. 46.

17. 21. 22. 34. B(I), I=1,...,N
0. 0. 0. 0.
0. 0. 0. 0. X(I), I=1,...,N

9.3 Example Results

F11BDF Example Program Results

Monitoring at iteration no. 1
residual norm: 1.4059E+02

Solution vector Residual vector
-4.5858E+00 1.5256E+01
1.0154E+00 2.6624E+01

-2.2234E+00 -8.7498E+00
6.0097E+00 1.8602E+01
1.3827E+00 8.2821E+00

-7.9070E+00 2.0416E+01
4.4270E-01 9.6094E+00
5.9248E+00 3.3055E+01

Monitoring at iteration no. 2
residual norm: 3.2742E+01

Solution vector Residual vector
4.1642E+00 -2.9585E+00
4.9370E+00 -5.5523E+00
4.8101E+00 8.2070E-01
5.4324E+00 -1.6828E+01
5.8531E+00 5.5975E-01
1.1925E+01 -1.9150E+00
8.4826E+00 1.0081E+00
6.0625E+00 -3.1004E+00

Final Results
Number of iterations for convergence: 3
Residual norm: 1.0373E-08
Right-hand side of termination criterion: 5.5900E-06
1-norm of matrix A: 1.1000E+01

Solution vector Residual vector
1.0000E+00 -1.3554E-09
2.0000E+00 -2.6109E-09
3.0000E+00 2.2471E-10
4.0000E+00 -3.2203E-09
5.0000E+00 6.3045E-10
6.0000E+00 -5.2431E-10
7.0000E+00 9.5771E-10
8.0000E+00 -8.4890E-10

